
capture log close

set more off
set matsize 10000


use rep_final, clear

	*noncitizens
	keep if cit==0
	*married last year, currently still married with spouse present
	keep if marrinyr==2 & married==1 
	*spouse info is avail.
	drop if citizen_sp==.

* Define control variables 

	local demog "age age2 male yreduc i.racegr"
	local imm "ysm i.bpl " 
	local fe "i.statefip i.yrmarr "
	
	local full "`demog' `imm' `fe' "

* Use yrmarr to measure time treatment because individuals are surveyed throughout the year

	g post1=(yrmarr>=2013 & yrmarr<=2016)
	g post2=(yrmarr>=2017)
	
	g treatpost1=treatgr*post1
	g treatpost2=treatgr*post2
	
	
	
	*by spouse's origin
		foreach controlgr of varlist controlgrdaca {
		local a=1
		foreach depvar of varlist native_sp citimm_sp {
	
			
			display ""
			display "--------------------------------------"
			
			display "Dependent Variable: `depvar' "
			display "Control group: `controlgr' "
			
			
			display "--------------------------------------"
			

			*treatpost1 and treatpost2 both compare to pre-period.

			
			reg `depvar' treatpost1 treatpost2 treatgr `full'  ///
				if (daca==1|`controlgr'==1)    [pweight = perwt] , cluster(statefip)
			est sto e`a'	
			
			local a=`a'+1
			
			*dep var mean
			sum `depvar' [fweight=perwt] if e(sample)
			
			}
		}
	

			*make table
			esttab e1 e2 ///
					using table4.csv, se  ///
					mtitles("Married to a Native Citizen" ///
							"Married to an Immigrant Citizen " ) ///
					title(Intermarriage Rate) ///
					b(%9.4f) se(%9.4f) star(* 0.1 ** 0.05 *** 0.01) ///
					nogaps replace		
